

ANALISI_PIANA_INTAGLIO_loop

```

FINISH
/CLEAR
C*****
C***
C*** USO DI ELEMENTI PIANI
C***
C*** - CORPO IN "PLANE STRAIN"
C***
C*** USO DI COMANDI DI PROGRAMMAZIONE
C***
*do,i,0,5,1
/PREP7
*if,i,gt,0,THEN
aclear,all
adele,all
ldele,all
kdele,all
*endif
D=30      ! semi-larghezza / raggio nominale
L=90      ! semi-lunghezza
RR=5+2*i  ! raggio di raccordo
NC=12     ! n° di divisioni nella zona del raccordo
ESZ1=2    ! dimensioni elementi
PA=10     ! tensione nominale in senso assiale
C***
C*** AREE
C***
RECTNG,0,D,0,D
RECTNG,0,D,D,L
AGLUE,1,2                ! Unisce le due aree rettangolari
CYL4,D,0,RR
ASEL,,LOC,X,D-0.1,D+0.1 ! trova il numero del cerchio
*GET,NA,AREA,,NUM,MAX
ASEL,ALL
ASBA,1,NA
C***
C*** MATERIALE
C***
MP,EX,1,210000
C***
C*** VINCOLI DI SIMMETRIA
C***
LSEL,,LOC,Y,-1,0.001
DL,ALL,,SYMM
LSEL,ALL
LSEL,,LOC,X,-1,0.001
DL,ALL,,SYMM
LSEL,ALL
C***
C*** CARICHI
C***
LSEL,,LOC,Y,L-0.001,L+1
SFL,ALL,PRESS,-PA
C***
C*** MESH
/TITLE,STATO PIANO DI DEFORMAZIONE
ET,1,182,, ,2
ESIZE,ESZ1                ! dimensioni elementi
LSEL,,LOC,X,-0.001,0.001 ! concatena i due lati opposti al raccordo
LSEL,R,LOC,Y,-1,D-0.01   ! in modo che l'area abbia 4 lati
*GET,NL1,LINE,,NUM,MAX
LSEL,,LOC,Y,D-0.001,D+0.01
*GET,NL2,LINE,,NUM,MAX
LSEL,ALL
LCCAT,NL1,NL2
MSHAPE,0                ! parametri per il mesh "mapped"
MSHKEY,1
AMESH,ALL

```

ANALISI_PIANA_INTAGLIO_1oop

```
C***  
C*** SOLUZIONE  
C***  
FINISH  
/SOLU  
SOLVE  
FINISH  
C***  
C*** POST-PROCESSING  
C***  
/POST1  
SET  
PLNSOL,S,Y  
*get,symax,plnso1,0,max  
RD=RR/(D-RR)  
k1=PA*D/(D-RR)  
kt=symax/k1  
/output,kt_vs_rd,txt,,append  
*status,rd  
*status,kt  
/output,,  
finish  
*enddo
```